@echo off
:: 设置控制台使用 UTF-8 编码
chcp 65001

:: 检查 dotnet 是否安装
dotnet --version >nul 2>&1
if %errorlevel% neq 0 (
    echo Dotnet SDK 未找到，请先安装 .NET SDK。
    pause
    exit /b 1
)

:: 检查是否已安装指定模板包
dotnet new --list | findstr "VerEasy.Core" >nul 2>&1
if %errorlevel% neq 0 (
    echo Vereasy 模板未安装，正在安装最新版本...
    dotnet new install VerEasyCoreTemplate >nul 2>&1
)

:: 提示用户输入项目名称
echo ===========================
echo 欢迎使用项目创建工具
echo ===========================
set /p PROJECT_NAME=请输入项目名称（例如 MyCore）： 

:: 提示用户输入输出路径
echo ===========================
set /p OUTPUT_PATH=请输入输出路径（留空则使用当前目录）： 

:: 如果未提供路径，则使用当前目录
if "%OUTPUT_PATH%"=="" set OUTPUT_PATH=%cd%

:: 显示创建项目信息
echo ===========================
echo 您即将创建的项目名称是：%PROJECT_NAME%
echo 项目输出路径为：%OUTPUT_PATH%
echo ===========================

:: 延迟 0.5 秒后开始创建
echo 正在创建项目 "%PROJECT_NAME%" 到路径 "%OUTPUT_PATH%"，请稍候...
timeout /nobreak /t 1 >nul

:: 创建项目
dotnet new vereasy --name "%PROJECT_NAME%" --output "%OUTPUT_PATH%\%PROJECT_NAME%"
if %errorlevel% neq 0 (
    echo 创建项目 "%PROJECT_NAME%" 失败，请检查错误信息。
    pause
    exit /b 1
)

echo 项目 "%PROJECT_NAME%" 创建成功，路径为 "%OUTPUT_PATH%\%PROJECT_NAME%"。
pause
